

```

PRBLNK EQU $F948
RDKEY EQU $FD0C
GETLINZ EQU $FD6F
OUTSP EQU $DB57
LINPRT EQU $ED24
COUT1 EQU $FDF0
COUTZ EQU $FDF6
CROUT EQU $FD8E
WAIT EQU $FCA8
BELL2 EQU $FBE4
COUT EQU $FDED
HOME EQU $FC58
MEM00 EQU $00
MEM01 EQU $01
MEM02 EQU $02
MEM03 EQU $03
MEM04 EQU $04
MEM05 EQU $05
MEM06 EQU $06
MEM07 EQU $07
INVFLG EQU $32
ORG $800

```

*

* DEBUT

*

```

ZEUS JSR HOME
      JSR DEMANDE
      LDA #$01
      STA PARAM
      JSR DISK
      JMP MENU1

```

*

* LECTURE.ECRITURE

*

```

DISK LDA #<IBDCTP
      LDY #>IBDCTP
      STA SIOB+6
      STY SIOB+7
      LDY #$10
L0    LDA SIOB,Y
      STA IOB,Y
      DEY
      BPL L0
      LDA PARAM
      STA IOB+12
L1    LDA #>IOB
      LDY #<IOB
      JSR $3D9
      LDA #$FF
      JSR WAIT
      DEC IOB+9
      DEC IOB+5
      BPL L1
      RTS

```

*

* TABLE.PARAMETRES

*

```

SIOB  HEX 01600100
      HEX 000F0000
      HEX 008F0000
      HEX 01000060
      HEX 01
IBDCTP HEX 0001EFD8
IOB    HEX 00000000
      HEX 00000000
      HEX 00000000
      HEX 00000000
      HEX 00
PARAM  HEX 00

```

*

* 1ERE PAGE

*

```

DEMANDE JSR HOME
         JSR BIP
         LDA #$3F
         STA INVFLG
         JSR XYPRINT

```

```

    HEX    040F
    ASC    "ZEUS"
    HEX    00
    JSR    XYPRINT
    HEX    080A
    ASC    "COPYRIGHT 1986"
    HEX    00
    JSR    MPRINT
    ASC    "DIDIER  VALLET"
    HEX    00
    LDA    #$FF
    STA    INVFLG
    JSR    XYPRINT
    HEX    1005
    ASC    "VEUILLEZ INTRODUIRE VOTRE"
    HEX    8D
    ASC    "          DISQUETTE DE BARD'S TALE"
    HEX    00
D0     JSR    CLAVIER
       CMP    #$A0
       BNE    D0
       RTS

```

*

* SONNERIE

*

```

BIP     JSR    BIP2
       LDA    #$40
       JSR    WAIT
BIP2    LDY    #$40
       JSR    BELL2
       RTS

```

*

* LECTURE DU CLAVIER

*

```

CLAVIER BIT    $C010
BCL     BIT    $C000
       BPL    BCL
       LDA    $C000
       BIT    $C010
       RTS

```

*

* PRINT CLASSIQUE

*

```

XYPRINT PLA
       STA    MEM00
       PLA
       STA    MEM01
       LDY    #$00
       INC    MEM00
       BNE    EC1
       INC    MEM01
EC1     LDA    ($00),Y
       STA    MEM02
       STA    $25
       JSR    CROUT
       INC    MEM00
       BNE    EC0
       INC    MEM01
EC0     LDA    ($00),Y
       STA    MEM03
       STA    $24
EC2     INC    MEM00
       BNE    EC3
       INC    MEM01
EC3     LDA    ($00),Y
       BEQ    EC4
       JSR    COUT
       JMP    EC2
EC4     LDA    MEM01
       PHA
       LDA    MEM00
       PHA
       RTS

```

*

* 2EME PRINT

*

```

MPRINT  JSR    CROUT

```

```

LDA MEM03
STA $24
PRINT PLA
STA MEM00
PLA
STA MEM01
LDY #$00
PR0 INC MEM00
BNE PR1
INC MEM01
PR1 LDA ($00),Y
BEQ PR2
JSR COUT
JMP PR0
PR2 LDA MEM01
PHA
LDA MEM00
PHA
RTS

*
* MENU PRINCIPAL
*
MENU1 JSR HOME
LDA #$3F
STA INVFLG
JSR XYPRINT
HEX 020A
ASC "MENU GENERAL - ZEUS"
HEX 00
LDA #$FF
STA INVFLG
JSR XYPRINT
HEX 0D02
ASC "1: Chargement de caracteres"
HEX 8D
ASC " 2: Catalogue des personnages."
HEX 8D
ASC " 3: Reviser un personnage."
HEX 8D
ASC " 4: Sauvegarder."
HEX 8D
ASC " 5: Undelete Personnages."
HEX 8D
ASC " 6: Quitter."
HEX 8D00
MN0 JSR CLAVIER
SEC
SBC #$B1
BMI MN0
CMP #$6
BPL MN0
ASL
CLC
ADC #<SAUT1
STA MN1+1
LDA #$00
ADC #>SAUT1
STA MN1+2
JSR HOME
MN1 JMP ($0000)
SAUT1 DA LOAD
DA CATALOG
DA REVISER
DA SAVE
DA UNDELETE
DA QUITTER

*
* SI TOUCHE 1 ENFONCEE
*
LOAD JMP ZEUS
*
* TOUCHE 2
*
CATALOG JSR CAT
JSR MAG1
JMP MENU1
CAT LDA #$3F

```

```

STA  INVFLG
JSR  PRINT
ASC  "   CATALOGUE DES PERSONNAGES PRESENTS   "
ASC  "           SUR CE DISK DE BARD'S TALE           "
HEX  8D8D8D00
LDA  #$FF
STA  INVFLG
LDA  #$01
STA  MEM04
LDA  #$10
STA  MEM05
LDA  #$81
STA  CA1+2
STA  CA5+2
STA  CA7+2
LDA  #$00
STA  CA1+1
STA  CA7+1
LDA  #$01
STA  CA5+1
LDA  #$88
STA  CA2+2
STA  CA11+2
STA  CA9+2
LDA  #$80
STA  CA2+1
STA  CA11+1
LDA  #$81
STA  CA9+1
CA0  LDA  MEM04
      CMP  #$0A
      BPL  CA3
      LDA  #$B0
CA3  JSR  COUTZ
      LDX  MEM04
      LDA  #$00
      JSR  LINPRT
      LDA  #$BA
      JSR  COUT1
      LDY  #$00
CA1  LDA  $8100,Y
      BNE  CA4
CA17 LDA  #$AE
CA4  CMP  #$FF
      BEQ  CA17
      JSR  COUTZ
      INY
      CPY  #$08
      BMI  CA1
      JSR  OUTSP
CA5  LDA  $8101
      CMP  #$00
      BEQ  CA6
CA7  LDA  $8100
      CMP  #$00
      BNE  CA6
      JSR  PRINT
      ASC  "Deleted"
      HEX  00
CA6  LDA  #20
      STA  $24
      LDA  #$00
      LDX  MEM05
      JSR  LINPRT
      LDA  #$BA
      JSR  COUTZ
      LDY  #$00
CA2  LDA  $8880,Y
      BNE  CA8
CA18 LDA  #$AE
CA8  CMP  #$FF
      BEQ  CA18
      JSR  COUTZ
      INY
      CPY  #$08
      BMI  CA2
      JSR  OUTSP

```

```

CA9      LDA    $8881
         BEQ    CA10
CA11     LDA    $8880
         BNE    CA10
         JSR    PRINT
         ASC    "Deleted"
         HEX    00
CA10     JSR    CROUT
         INC    MEM04
         INC    MEM05
         CLC
         LDA    CA1+1
         ADC    #$80
         STA    CA1+1
         STA    CA7+1
         BCC    CA13
         INC    CA1+2
         INC    CA7+2
CA13     CLC
         LDA    CA5+1
         ADC    #$80
         BCC    CA15
         INC    CA5+2
CA15     CLC
         LDA    CA2+1
         ADC    #$80
         STA    CA2+1
         STA    CA11+1
         BCC    CA14
         INC    CA2+2
         INC    CA11+2
CA14     LDA    CA9+1
         ADC    #$80
         STA    CA9+1
         BCC    CA16
         INC    CA9+2
CA16     LDA    MEM04
         CMP    #$10
         BPL    CA12
         JMP    PRINT
CA12     JSR    PRINT
         HEX    8D8D00
         RTS
*
* TOUCHE 3
*
REVISER  JSR    CAT
         JSR    PRINT
         ASC    " PERSONNAGE CHOISI: "
         HEX    00
         JSR    GETLINZ
         LDA    #$81
         STA    RE1+2
         LDA    #$00
         STA    RE1+1
RE2      LDY    #$FF
RE3      INY
         LDA    $200,Y
         CMP    #$8D
         BEQ    RE5
RE1      CMP    $1234,Y
         BEQ    RE3
         CLC
         LDA    RE1+1
         ADC    #$80
         STA    RE1+1
         BCC    RE4
         INC    RE1+2
         LDA    RE1+2
         CMP    #$90
         BNE    RE4
         JMP    RE6
RE4      JMP    RE2
RE5      CPY    #$00
         BNE    RE8
         JMP    MENU1
RE8      LDA    RE1+1

```

```

        STA  MEM06
        LDA  RE1+2
        STA  MEM07
RE7     JSR  MENU2
RE0     JSR  CLAVIER
        SEC
        SBC  #$C1
        BMI  RE0
        CMP  #$12
        BPL  RE0
        ASL
        CLC
        ADC  #<SAUT2
        STA  MN2+1
        LDA  #$00
        ADC  #>SAUT2
        STA  MN2+2
        JSR  HOME
MN2     JMP  ($0000)
*
*  PERSONNAGE NON EXISTANT
*
RE6     JSR  HOME
        LDA  #$3F
        STA  INVFLG
        JSR  XYPRINT
        HEX  0800
        ASC  "          CE PERSONNAGE N'EXISTE PAS          "
        ASC  "          SUR CETTE DISQUETTE SCENARIO        "
        ASC  "          VEUILLEZ PRESSER UNE TOUCHE         "
        HEX  00
        LDA  #$FF
        STA  INVFLG
        JSR  CLAVIER
        JMP  MENU1
*
*  TABLE DU 2 EME MENU
*
SAUT2   DA  DISPLAY
        DA  HEALING
        DA  MAGASIN
        DA  HPOINTS
        DA  QUALITES
        DA  ROUND
        DA  GOLD
        DA  EXPERT
        DA  LEVEL
        DA  NAME
        DA  RACE
        DA  STATUS
        DA  HPMAX
        DA  CLASS
        DA  SPELL
        DA  SPPTS
        DA  SPMAX
        DA  MENU1
*
*  SAUVEGARDE ET CALCUL
*
SAVE    LDA  #$02
        STA  PARAM
        LDX  #30
        LDA  #$7F
        STA  SA0+1
        STA  SA2+1
        LDA  #$81
        STA  SA0+2
        STA  SA2+2
        LDA  #$FF
        STA  SA1+1
        LDA  #$80
        STA  SA1+2
SA0     LDA  $1234
        LDY  #$80
SA1     EOR  $1234,Y
        DEY
        BNE SA1

```

```

SA2      STA      $1234
         LDA      SA0+1
         STA      SA1+1
         CLC
         ADC      #$80
         STA      SA0+1
         STA      SA2+1
         BCC      SA3
         INC      SA0+2
         INC      SA2+2
         JMP      SA4
SA3      INC      SA1+2
SA4      DEX
         BNE      SA0
         LDA      #$3F
         STA      INVFLG
         JSR      XYPRINT
         HEX      0A00
         ASC      "          SAUVEGARDE DES PERSONNAGES      "
         ASC      "          PRESSEZ UNE TOUCHE              "
         ASC      "          DEL POUR ANNULER                "
         HEX      00
         LDA      #$FF
         STA      INVFLG
         JSR      CLAVIER
         CMP      #$FF
         BEQ      SA5
         JSR      DISK
SA5      JMP      MENU1
*
* AJOUT D'UN Z POUR LES UNDELETE
*
UNDELETE LDA      #$00
         STA      UN0+1
         STA      UN1+1
         LDA      #$81
         STA      UN0+2
         STA      UN1+2
         LDY      #30
UN0      LDA      $1234
         BNE      UN1
UN1      LDA      #$DA
         STA      $1234
         LDA      UN0+1
         CLC
         ADC      #$80
         STA      UN0+1
         STA      UN1+1
         BCC      UN2
         INC      UN0+2
         INC      UN1+2
UN2      DEY
         BNE      UN0
         JMP      CATALOG
*
* FIN ET RETOUR AU JEU
*
QUITTER JSR      HOME
         LDA      #$3F
         STA      INVFLG
         JSR      PRINT
         ASC      "          INSEREZ UNE AUTRE DISQUETTE      "
         HEX      00
         LDA      #$FF
         STA      INVFLG
         JSR      CLAVIER
         JMP      $C600
*
* MENU SECONDAIRE
*
MENU2    JSR      HOME
         LDA      #$3F
         STA      INVFLG
         JSR      PRINT
         ASC      "          MODIFICATION DE PERSONNAGES      "
         HEX      00
         LDA      #$FF

```

```

STA   INVFLG
JSR   PRINT
HEX   8D
ASC   "   A: Affiche le caractere."
HEX   8D
ASC   "   B: Bard's plus healing."
HEX   8D
ASC   "   C: Liste des sorts."
HEX   8D
ASC   "   D: Modifier les points de vie."
HEX   8D
ASC   "   E: Modifier les qualites."
HEX   8D
ASC   "   F: Nombre d'attaques par round."
HEX   8D
ASC   "   G: Nombre de pieces d'or."
HEX   8D
ASC   "   H: Nombre de points d'experience."
HEX   8D
ASC   "   I: Changement du niveau."
HEX   8D
ASC   "   J: Changer le nom du personnage."
HEX   8D
ASC   "   K: Changer de race."
HEX   8D
ASC   "   L: Changer d'etat."
HEX   8D
ASC   "   M: Restore les points de vie."
HEX   8D
ASC   "   N: Changer de classe."
HEX   8D
ASC   "   O: Ajouter des sorts."
HEX   8D
ASC   "   P: Nombre de Spellpoints maxi."
HEX   8D
ASC   "   Q: Restore Spellpoints."
HEX   8D
ASC   "   R: Retour au menu general."
HEX   8D8D00
LDA   #$3F
STA   INVFLG
JSR   PRINT
ASC   " PRESSER LA TOUCHE DE L'OPTION CHOISIE  "
HEX   00
LDA   #$FF
STA   INVFLG
RTS

```

```

*
* AFFICHE LES CARACTERISTIQUES
*

```

```

DISPLAY LDA   #$3F
        STA   INVFLG
        JSR   XYPRINT
        HEX   0107
        ASC   "PERSONNAGE: "
        HEX   00
        LDA   #$FF
        STA   INVFLG
        JSR   PRBLNK
        LDY   #$FF
DI1     INY
        CPY   #$10
        BEQ   DI2
        LDA   ($06),Y
        CMP   #$FF
        BEQ   DI2
        JSR   COUTZ
        JMP   DI1
DI2     JSR   CROUT
        JSR   CROUT
        LDY   #$10
        LDX   #$00
        JSR   DECODE
        JSR   DECODE
        LDA   #$04
        STA   MEM03
        STA   $24

```



```
LDA #20
STA MEM04
JSR PRINT
ASC "Force....: "
HEX 00
LDY #$00
STY DE1
JSR DI3
JSR PRINT
ASC "Intelligence: "
HEX 00
JSR DI3
JSR PRINT
ASC "Dexterite: "
HEX 00
JSR DI3
JSR PRINT
ASC "Constitution: "
HEX 00
JSR DI3
JSR PRINT
ASC "Luck.....: "
HEX 00
JSR DI3
JSR PRINT
ASC "Inconnue....: "
HEX 00
JSR DI3
JSR CROUT
JSR MPRINT
ASC "Experience: "
HEX 00
LDA #$20
STA MEM04
LDY #$14
JSR ROUT1
JSR MPRINT
ASC "Or.....: "
HEX 00
LDA #$30
STA MEM04
LDY #$24
JSR ROUT1
JSR CROUT
JSR CROUT
LDA #15
STA MEM03
STA $24
JSR PRINT
ASC "Niveau maxi...: "
HEX 00
LDY #$21
JSR DI5
JSR PRINT
ASC "Niveau actuel.: "
HEX 00
LDY #$23
JSR DI5
JSR PRINT
ASC "Points maxi...: "
HEX 00
LDY #$31
JSR DI5
JSR PRINT
ASC "Points actuels: "
HEX 00
LDY #$33
JSR DI5
JSR PRINT
ASC "SpPts maxi....: "
HEX 00
LDY #$35
JSR DI5
JSR PRINT
ASC "SpPts actuels.: "
HEX 00
LDY #$37
```

```

        JSR    DI5
        JSR    CROUT
        JSR    MAG1
        JMP    RE7
DI5     LDA    ($06),Y
        TAX
        DEY
        LDA    ($06),Y
        JSR    LINPRT
        JSR    CROUT
        LDA    MEM03
        STA    $24
        RTS
DI3     LDY    DE1
        LDA    DE0,Y
        TAX
        LDA    #$00
        JSR    LINPRT
        INC    DE1
        LDA    DE1
        AND    #$01
        BNE    DI4
        JSR    CROUT
        LDA    MEM03
        STA    $24
        RTS
DI4     LDA    MEM04
        STA    $24
        RTS
*
* SOINS INTENSIFS
*
HEALING LDY    #$20
        JSR    MAXI
        LDY    #$30
        JSR    MAXI
        LDY    #$34
        JSR    MAXI
        LDA    #$00
        LDY    #$3E
        STA    ($06),Y
        LDY    #$10
        LDA    ($06),Y
        CMP    #$08
        BNE    HEAL1
        LDY    #$63
        LDX    #$03
HEAL2   LDA    ($06),Y
        STA    HEAL0,X
        DEY
        DEX
        BPL    HEAL2
        LDX    #$03
        LDY    #$13
HEAL3   LDA    HEAL0,X
        STA    ($06),Y
        DEY
        DEX
        BPL    HEAL3
HEAL1   JMP    DISPLAY
HEAL0   HEX    00000000
*
* LISTE DE SORTS
*
MAGASIN LDA    #$3F
        STA    INVFLG
        JSR    PRINT
        ASC    "          LISTE DES SORTS DISPONIBLES          "
        HEX    8D00
        LDA    #$FF
        STA    INVFLG
        JSR    PRINT
        ASC    "LEV. CONJURER SORCERER MAGICIAN WIZARDRY"
        HEX    8D8D
        ASC    " 01      MAFL      MIJA      VOPL      SUDE"
        HEX    8D
        ASC    "          ARFI      PHBL      AIAR      REDE"

```

```

    HEX 8D
    ASC "          SOSH      LOTR      STLI "
    HEX 8D
    ASC "          TRZP      HYIM      SCSI "
    HEX 8D8D
    ASC " 02      FRFO      DISB      HOWA      LESU "
    HEX 8D
    ASC "          MACO      TADU      WIST      DEBA "
    HEX 8D
    ASC "          BASK      MIFI      MAGA "
    HEX 8D
    ASC "          WOHL      FEAR      AREN "
    HEX 8D8D
    ASC " 03      MAST      WIWO      MYSH      SUPH "
    HEX 8D
    ASC "          LERE      VANI      OGST      DISP "
    HEX 8D
    ASC "          LEVI      SESI      MIMI "
    HEX 8D
    ASC "          WAST      CURS      STFL "
    HEX 8D8D8D00
    JSR MAG1
    JSR PRINT
    ASC "LEV.  CONJURER  SORCERER  MAGICIAN  WIZARDRY"
    HEX 8D8D
    ASC " 04      INWD      CAEY      SPTO      PRSU "
    HEX 8D
    ASC "          FLRE      WIWA      DRBR      ANDE "
    HEX 8D
    ASC "          POST      INVI      STSI "
    HEX 8D8D
    ASC " 05      GRRE      WIOG      ANMA      SPBI "
    HEX 8D
    ASC "          WROV      DIIL      ANSW      DMST "
    HEX 8D
    ASC "          SHSP      MIBL      STTO "
    HEX 8D8D
    ASC " 06      INOG      WIDR      PHDO      SPSP "
    HEX 8D
    ASC "          MALE      MIWP      YMCA      BEDE "
    HEX 8D8D
    ASC " 07      FLAN      WIGI      FEST      GRSU "
    HEX 8D
    ASC "          AFAR      SOSI      DEST "
    HEX 8D8D8D00
    JSR MAG1
    JMP RE7
MAG1  LDA  #$3F
    STA  INVFLG
    JSR  PRINT
    ASC  " APPUYER SUR UNE  TOUCHE POUR CONTINUER  "
    HEX 00
    LDA  #$FF
    STA  INVFLG
    JSR  CLAVIER
    JSR  HOME
    RTS

*
* MODIFICATION DES HP
*
HPOINTS  LDA  #$3F
    STA  INVFLG
    JSR  FLECHE
    JSR  XYPRINT
    HEX 0304
    ASC  "CHANGEMENT DES POINTS DE VIE"
    HEX 8D8D8D00
COMMUN  LDY  #$31
    STY  DE1
    LDA  #$FF
    STA  INVFLG
    JSR  PRINT
    ASC  "          ACTUELS:  "
    HEX 00
    LDY  DE1
    LDA  ($06),Y
    DEY

```

```

TAX
LDA ($06),Y
JSR LINPRT
JSR PRINT
HEX 8D
ASC " DESIRES: "
HEX 00
JSR HEXA
LDA HE0
LDY DE1
STA ($06),Y
INY
INY
STA ($06),Y
DEY
LDA HE1
STA ($06),Y
DEY
DEY
STA ($06),Y
JSR PAUSE
JMP RE7

```

*

* MODIFICATION DES QUALITES

*

```

QUALITES LDY #$10
LDX #$00
JSR DECODE
JSR DECODE
LDA #$3F
STA INVFLG
JSR XYPRINT
HEX 0209
ASC "CHANGEMENT DES ABILITES"
HEX 8D00
LDA #$FF
STA INVFLG
JSR XYPRINT
HEX 030B
ASC "( 31 POINTS MAXIMUM) "
HEX 8D8D00
JSR PRINT
ASC "St. FORCE.....: "
HEX 8D
ASC "Iq. INTELLIGENCE:"
HEX 8D
ASC "Dx. DEXTERITE...:"
HEX 8D
ASC "Cn. CONSTITUTION:"
HEX 8D
ASC "Lk. CHANCE.....:"
HEX 8D
ASC "Xx. ....:"
HEX 8D00
LDA #$05
STA $25
LDA #18
STA MEM03
JSR CROUT
LDA #$00
STA DE1
QU0 LDA MEM03
STA $24
LDY DE1
LDA DE0,Y
CMP #10
BPL QU1
LDA #$B0
JSR COUTZ
LDA DE0,Y
QU1 INY
STY DE1
TAX
LDA #$00
JSR LINPRT
JSR CROUT
LDA DE1

```

```

      CMP    #$06
      BMI    QU0
      LDA    #$5
      STA    $25
      LDA    #$00
      STA    DE1
QU2   JSR    CROUT
QU3   LDA    #21
      STA    $24
      JSR    PRINT
      ASC    "DEVIENT:      "
      HEX    00
      LDA    #30
      STA    $24
      JSR    RDKEY
      CMP    #$B0
      BMI    QU3
      CMP    #$B4
      BPL    QU3
      JSR    COUTZ
      AND    #%00000011
      TAY
      LDA    CO0,Y
      LDX    DE1
      STA    CO1,X
      JSR    RDKEY
      CMP    #$B0
      BMI    QU3
      CMP    #$BA
      BPL    QU3
      JSR    COUTZ
      SEC
      SBC    #$B0
      LDX    DE1
      CLC
      ADC    CO1,X
      CMP    #32
      BPL    QU3
      STA    CO1,X
      INX
      CPX    #$06
      BPL    QU4
      STX    DE1
QU4   JMP    QU2
      LDY    #$10
      LDX    #$00
      JSR    CODE
      JSR    CODE
      JMP    RE7
CO0   HEX    000A141E
CODE  LDA    CO1,X
      ASL
      ASL
      ASL
      STA    DE1
      INX
      LDA    CO1,X
      LSR
      LSR
      ORA    DE1
      STA    ($06),Y
      INY
      LDA    CO1,X
      ASL
      ASL
      ASL
      ASL
      ASL
      ASL
      INX
      ORA    CO1,X
      STA    ($06),Y
      INY
      INX
      RTS
DECODE LDA    ($06),Y
      AND    #%11111000

```

```

LSR
LSR
LSR
STA    DE0,X
INX
LDA    ($06),Y
AND    #%00000111
ASL
ASL
STA    DE1
INX
LDA    ($06),Y
AND    #%11000000
CLC
ROL
ROL
ROL
ORA    DE1
STA    DE0,X
INX
LDA    ($06),Y
AND    #%00011111
STA    DE0,X
INX
INX
RTS
DE1    HEX    00
DE0    HEX    000000000000
CO1    HEX    000000000000
*
* ATTAQUES PAR ROUND
*
ROUND  LDA    #$3F
        STA    INVFLG
        JSR    FLECHE
        JSR    XYPRINT
        HEX    0305
        ASC    "NOMBRE D'ATTAQUES PAR ROUND"
        HEX    8D8D8D00
        LDA    #$FF
        STA    INVFLG
        JSR    PRINT
        ASC    "          ACTUEL:  "
        HEX    00
        LDY    #$4F
        STY    DE1
        LDA    ($06),Y
        TAX
        LDA    #$00
        JSR    LINPRT
        JSR    PRINT
        HEX    8D
        ASC    "          DESIRE:  "
        HEX    00
        JMP    COMM2
*
* PIECES D'OR POSSEDEES
*
GOLD   LDA    #$3F
        STA    INVFLG
        JSR    XYPRINT
        HEX    040B
        ASC    "NOMBRE DE PIECES D'OR"
        HEX    8D8D8D00
        LDA    #$FF
        STA    INVFLG
        JSR    PRINT
        ASC    "  PIECES D'OR EN BOURSE:  "
        HEX    00
        LDA    #$30
        STA    MEM04
        LDY    #$24
        JSR    ROUT1
        JSR    PRINT
        HEX    8D
        ASC    "  PIECES D'OR DEMANDEES:  "
        HEX    00

```

```

LDA #25
STA MEM03
LDA #24
STA MEM05
LDA #2F
STA MEM04
JSR ROUT2
JMP RE7
ROUT2 LDA MEM03
STA $24
LDA #00
STA DE1
JSR PRINT
ASC " "
HEX 00
LDA MEM03
STA $24
LDA #00
LDY #11
RO2 STA DE0,Y
DEY
BPL RO2
RO3 JSR RDKEY
CMP #8D
BEQ RO4
CMP #B0
BMI ROUT2
CMP #BA
BPL ROUT2
JSR COUTZ
SEC
SBC #B0
LDY DE1
STA DE0,Y
INY
STY DE1
CPY #12
BMI RO3
RO4 LDY MEM04
LDX DE1
DEX
RO5 LDA DE0,X
STA ($06),Y
DEY
DEX
BPL RO5
RO6 CPY MEM05
BMI RO7
LDA #00
STA ($06),Y
DEY
JMP RO6
RO7 RTS
ROUT1 LDA ($06),Y
CLC
ADC #B0
JSR COUTZ
INY
CPY MEM04
BMI ROUT1
RTS

*
* AJOUT D'EXPERIENCE
*
EXPERT LDA #3F
STA INVFLG
JSR XYPRINT
HEX 0603
ASC "NOMBRE DE POINTS D'EXPERIENCE"
HEX 8D8D8D00
LDA #FF
STA INVFLG
JSR PRINT
ASC " EXPERIENCE ACQUISE: "
HEX 00
LDA #20
STA MEM04

```

```

LDY    #$14
JSR    ROUT1
JSR    PRINT
HEX    8D
ASC    "  EXPERIENCE VOULUE. : "
HEX    00
LDA    #22
STA    MEM03
LDA    #$14
STA    MEM05
LDA    #$1F
STA    MEM04
JSR    ROUT2
JMP    RE7

```

*

* CHANGEMENT DE NIVEAU

*

```

LEVEL  LDA    #$3F
        STA    INVFLG
        JSR    FLECHE
        JSR    XYPRINT
        HEX    0308
        ASC    "NIVEAU DU PERSONNAGE"
        HEX    8D8D8D00
        LDA    #$FF
        STA    INVFLG
        LDY    #$21
        JMP    COMMUN

```

*

* CHANGEMENT DE NOM

*

```

NAME    LDA    #$3F
        STA    INVFLG
        JSR    XYPRINT
        HEX    020B
        ASC    "CHANGEMENT DE NOM"
        HEX    8D00
        LDA    #$FF
        STA    INVFLG
        JSR    XYPRINT
        HEX    0309
        ASC    "(16 CARACTERES MAXI.)"
        HEX    00
        JSR    PRINT
        HEX    8D8D
        ASC    "ANCIEN  NOM:  "
        HEX    00
        LDY    #$00
NA1     LDA    ($06),Y
        CMP    #$FF
        BNE    NA2
        LDA    #$AE
NA2     JSR    COUTZ
        INY
        CPY    #$10
        BMI    NA1
        JSR    PRINT
        HEX    8D
        ASC    "NOUVEAU NOM:  "
        HEX    00
        JSR    GETLINZ
        LDY    #$00
NA4     LDA    $200,Y
        CMP    #$8D
        BEQ    NA3
        STA    ($06),Y
        INY
        CPY    #$10
        BMI    NA4
        JMP    RE7
NA3     CPY    #$00
        BEQ    NA7
        LDA    #$FF
NA6     STA    ($06),Y
        INY
        CPY    #$10
        BMI    NA6

```



```
NA7      JMP      RE7
*
*  CHANGEMENT DE RACE
*
RACE     LDA      #$3F
         STA      INVFLG
         JSR      XYPRINT
         HEX      030A
         ASC      "CHANGEMENT DE LA RACE"
         HEX      8D8D00
         LDA      #$FF
         STA      INVFLG
         JSR      PRINT
RA3      ASC      "Races existantes: "
         HEX      00
         LDA      #18
         STA      MEM03
         DEC      $25
         JSR      MPRINT
         ASC      "0.Human (Humain)"
         HEX      00
         JSR      MPRINT
         ASC      "1.Elf (Elfe)"
         HEX      00
         JSR      MPRINT
         ASC      "2.Dwarf (Nain)"
         HEX      00
         JSR      MPRINT
         ASC      "3.Hobbit"
         HEX      00
         JSR      MPRINT
         ASC      "4.1/2 Elf"
         HEX      00
         JSR      MPRINT
         ASC      "5.1/2 Orc"
         HEX      00
         JSR      MPRINT
         ASC      "6.Gnome"
         HEX      8D8D8D00
         JSR      PRINT
         ASC      "Race actuelle: "
         HEX      00
         LDY      #$39
         LDA      ($06),Y
         BMI      RA1
         CMP      #$07
         BPL      RA1
         CLC
         ADC      #$B0
         LDY      #$FF
RA4      INY
         CMP      RA3,Y
         BNE      RA4
         INY
RA5      INY
         LDA      RA3,Y
         BEQ      RA6
         JSR      COUTZ
         JMP      RA5
RA6      JSR      CROUT
         JSR      PRINT
         ASC      "Race desiree.: "
         HEX      00
RA2      JSR      RDKEY
         CMP      #$8D
         BEQ      RA7
         CMP      #$B0
         BMI      RA2
         CMP      #$B7
         BPL      RA2
         JSR      COUTZ
         SEC
         SBC      #$B0
         LDY      #$39
         STA      ($06),Y
         JSR      PAUSE
RA7      JMP      RE7
```

```

RA1      JMP      ST1
*
*  CHANGEMENT DE L'ETAT PHYSIQUE
*
STATUS   LDA      #$3F
          STA      INVFLG
          JSR      XYPRINT
          HEX      030B
          ASC      "ETAT PHYSIQUE"
          HEX      8D00
          JSR      MPRINT
          ASC      "DU PERSONNAGE"
          HEX      8D8D8D00
          LDA      #$FF
          STA      INVFLG
          JSR      PRINT
STATE    ASC      "0.      : Etat normal."
          HEX      8D
          ASC      "1.Psnd: Empoisonnement."
          HEX      8D
          ASC      "2.Old : Vieillesse."
          HEX      8D
          ASC      "3.Dead: Mort."
          HEX      8D
          ASC      "4.Ston: Petrifie."
          HEX      8D
          ASC      "5.Para: Paralysation."
          HEX      8D
          ASC      "6.Poss: Possede."
          HEX      8D
          ASC      "7.Nuts: Folie."
          HEX      8D8D00
          JSR      PRINT
          ASC      "      ETAT ACTUEL: "
          HEX      00
          LDY      #$3E
          LDA      ($06),Y
          CLC
          ADC      #$B0
          LDY      #$FF
ST0      INY
          CPY      #$FF
          BEQ      ST1
          CMP      STATE,Y
          BNE      ST0
          INY
ST2      INY
          LDA      STATE,Y
          CMP      #$BA
          BEQ      ST3
          JSR      COUTZ
          JMP      ST2
ST3      JSR      PRINT
          HEX      8D
          ASC      "      ETAT DESIRE: "
          HEX      00
ST4      JSR      RDKEY
          CMP      #$8D
          BEQ      ST5
          CMP      #$B0
          BMI      ST4
          CMP      #$B8
          BPL      ST4
          JSR      COUTZ
          LDY      #$3E
          SEC
          SBC      #$B0
          STA      ($06),Y
          JSR      PAUSE
ST5      JMP      RE7
ST1      JSR      HOME
          LDA      #$3F
          STA      INVFLG
          JSR      XYPRINT
          HEX      0700
          ASC      "
          ASC      "      PERSONNAGE ERRONE      "
          ASC      "      TRAITEMENT IMPOSSIBLE      "

```

```

    HEX    00
    LDA    #$FF
    STA    INVFLG
    JSR    CLAVIER
    JMP    RE7
HPMAX   LDY    #$30
HP1     JSR    MAXI
        JSR    MAX1
        JMP    RE7

*
* FAIRE 36 METIERS
*
CLASS   LDA    #$3F
        STA    INVFLG
        JSR    XYPRINT
        HEX    030A
        ASC    "CHANGEMENT DU METIER"
        HEX    8D8D8D00
        LDA    #$FF
        STA    INVFLG
        JSR    PRINT
CL0     ASC    "Possibilites: 0.Warrior (Guerrier)"
        HEX    8D
        ASC    "          1.Wizardry"
        HEX    8D
        ASC    "          2.Sorcerer"
        HEX    8D
        ASC    "          3.Conjurer (Clerc)"
        HEX    8D
        ASC    "          4.Magicien"
        HEX    8D
        ASC    "          5.Rogue (Voleur)"
        HEX    8D
        ASC    "          6.Barde"
        HEX    8D
        ASC    "          7.Paladin"
        HEX    8D
        ASC    "          8.Hunter (Chasseur)"
        HEX    8D
        ASC    "          9.Monk (Moine)"
CL6     HEX    8D8D8D
        ASC    "  TALENT ACTUEL: "
        HEX    00
        LDY    #$38
        LDA    ($06),Y
        PHA
        BMI    CL1
        CMP    #$0A
        BPL    CL1
        CMP    #$09
        BNE    CL8
        LDA    #>CL6
        STA    CL10+2
        STA    CL11+2
        LDA    #<CL6
        STA    CL10+1
        STA    CL11+1
        JMP    CL9
CL8     LDA    #>CL0
        STA    CL10+2
        STA    CL11+2
        LDA    #<CL0
        STA    CL10+1
        STA    CL11+1
CL9     LDY    #$FF
        PLA
        CLC
        ADC    #$B0
CL2     INY
CL10    CMP    CL0,Y
        BNE    CL2
        INY
CL3     INY
CL11    LDA    CL0,Y
        JSR    COUTZ
        CMP    #$8D
        BNE    CL3

```

```

      JSR   PRINT
      ASC   "    TALENT DESIRE:  "
      HEX   00
CL4    JSR   RDKEY
      CMP   #$8D
      BEQ   CL5
      CMP   #$B0
      BMI   CL4
      CMP   #$BA
      BPL   CL4
      JSR   COUTZ
      SEC
      SBC   #$B0
      LDY   #$38
      STA   ($06),Y
      JSR   PAUSE
CL5    JMP   RE7
CL1    JMP   ST1
*
* ACQUERIR DE NOUVEAUX SORTS
*
SPELL  LDA   #$3F
      STA   INVFLG
      JSR   FLECHE
      JSR   XYPRINT
      HEX   0304
      ASC   "CHANGEMENT DE LA LISTE DE SORTS"
      HEX   8D8D8D00
      LDA   #$FF
      STA   INVFLG
      JSR   PRINT
      ASC   "NIVEAU ACQUIS EN  1.SORCERER:  "
      HEX   00
      LDA   #18
      STA   MEM03
      LDY   #$40
      JSR   SE0
      JSR   MPRINT
      ASC   "2.CONJURER:  "
      HEX   00
      LDY   #$41
      JSR   SE0
      JSR   MPRINT
      ASC   "3.MAGICIEN:  "
      HEX   00
      LDY   #$42
      JSR   SE0
      JSR   MPRINT
      ASC   "4.WIZARDRY:  "
      HEX   00
      LDY   #$43
      JSR   SE0
      JSR   PRINT
      HEX   8D8D
      ASC   "QUE DESIREZ-VOUS MODIFIER:  "
      HEX   00
SE2    JSR   RDKEY
      CMP   #$8D
      BEQ   SE1
      CMP   #$B1
      BMI   SE2
      CMP   #$B5
      BPL   SE2
      JSR   COUTZ
      SEC
      SBC   #$71
      STA   DE1
      JSR   PRINT
      HEX   8D
      ASC   "POUR QUEL NIVEAU.....:  "
      HEX   00
COMM2  LDA   #$01
      STA   FLAG
      JSR   HEXA
      LDA   HE0
      LDY   DE1
      STA   ($06),Y

```

```

    JSR   PAUSE
SE1     JMP   RE7
SE0     LDA   ($06),Y
        TAX
        LDA   #$00
        JSR   LINPRT
        RTS

*
* AUGMENTER SON POUVOIR
*
SPPTS   LDA   #$3F
        STA  INVFLG
        JSR   FLECHE
        JSR   XYPRINT
        HEX  0304
        ASC  "NOMBRE DE POINTS DE MAGIE"
        HEX  8D8D8D00
        LDY  #$35
        JMP  COMMUN
SPMAX   LDY  #$34
        JMP  HP1
MAXI    LDA   ($06),Y
        PHA
        INY
        LDA   ($06),Y
        INY
        INY
        STA  ($06),Y
        DEY
        PLA
        STA  ($06),Y
        RTS
MAX1    LDA   #$3F
        STA  INVFLG
        JSR   XYPRINT
        HEX  070A
        ASC  "RESTAURATION EFFECTUEE"
        HEX  00
        JSR   MPRINT
        ASC  " PRESSEZ UNE TOUCHE  "
        HEX  00
        LDA  #$FF
        STA  INVFLG
        JSR   CLAVIER
        RTS
PAUSE   LDX  #$80
PA1     LDY  #$FF
PA2     DEY
        BNE  PA2
        DEX
        BNE  PA1
        RTS
HEXA    LDA  #$00
        STA  HE1
        STA  HE0
        LDA  $24
HE99    STA  MEM03
        LDA  MEM03
        STA  $24
        LDA  HE1
        LDX  HE0
        JSR  LINPRT
        JSR  PRINT
        ASC  "      "
        HEX  00
        LDX  #$10
HE2     JSR  PA1
        JSR  CLAVIER
        CMP  #$8D
        BEQ  HE3
        CMP  #$88
        BEQ  HE4
        CMP  #$95
        BNE  HE2
        INC  HE0
        BNE  HE10
        LDA  FLAG

```

```
BNE HE10
INC HE1
HE10 JMP HE99
HE4 DEC HE0
      BNE HE10
      LDA FLAG
      BNE HE10
      DEC HE1
      JMP HE99
HE3 LDA #$00
     STA FLAG
     RTS
HE0 HEX 00
HE1 HEX 00
FLAG HEX 00
FLECHE JSR XYPRINT
        HEX 1508
        ASC "UTILISEZ LES FLECHES"
        HEX 00
        RTS
```